Turn On/Off Functions to speed things up I have these two macros that are like my Yin &amp; Yang. They turn off and on functions in excel so my macros calculate quicker.

I call on in the beginning and one at the end.

Is there anything I can do or add to these to make them better?

Private Sub TurnOffFunctions()

    Application.ScreenUpdating = False
    Application.DisplayStatusBar = False
    Application.Calculation = xlCalculationManual
    Application.EnableEvents = False

End Sub

Private Sub TurnOnFunctions()

    Application.ScreenUpdating = True
    Application.DisplayStatusBar = True
    Application.Calculation = xlCalculationAutomatic
    Application.EnableEvents = True

End Sub
You should save the initial state of the Application.* variables before you mess with them, then reset them when you're done. Not all of your users will have the same settings and it's rude to assume they do. 

A simple way to do it would be to implement a class like this:

'@Folder("Classes")
Option Explicit

Private Type ApplicationSettings
  ScreenUpdating As Boolean
  DisplayStatusBar As Boolean
  CalculationMethod As Excel.XlCalculation
  EnableEvents As Boolean
End Type

Private this As ApplicationSettings

Public Property Get ScreenUpdating() As Boolean
  ScreenUpdating = this.ScreenUpdating
End Property

Public Property Get DisplayStatusBar() As Boolean
  DisplayStatusBar = this.DisplayStatusBar
End Property

Public Property Get CalculationMethod() As Excel.XlCalculation
  CalculationMethod = this.CalculationMethod
End Property

Public Function DisableExcelDisplayUpdates()

  SetScreenUpdating False
  SetDisplayStatusBar False
  SetCalculationMethod Excel.xlCalculationManual
  SetEnableEvents False

End Function

Public Function EnableExcelDisplayUpdates()

  Application.ScreenUpdating = this.ScreenUpdating
  Application.DisplayStatusBar = this.DisplayStatusBar
  Application.Calculation = this.CalculationMethod
  Application.EnableEvents = this.EnableEvents

End Function

Private Function SetScreenUpdating(ByVal value As Boolean)
  this.ScreenUpdating = Application.ScreenUpdating
  Application.ScreenUpdating = value
End Function

Private Function SetDisplayStatusBar(ByVal value As Boolean)
  this.DisplayStatusBar = Application.DisplayStatusBar
  Application.DisplayStatusBar = value
End Function

Private Function SetCalculationMethod(ByVal value As Excel.XlCalculation)
  this.CalculationMethod = Application.Calculation
  Application.Calculation = value
End Function

Private Function SetEnableEvents(ByVal value As Boolean)
  this.EnableEvents = Application.EnableEvents
  Application.EnableEvents = value
End Function


Then test it out like this:

Option Explicit
`@Folder("Tests")

Public Sub testIt()

  Dim ExcelValues As Class1
  Set ExcelValues = New Class1

  Debug.Print "Before "
  Debug.Print "ScreenUpdating: " &amp; Application.ScreenUpdating
  Debug.Print "DisplayStatusBar: " &amp; Application.DisplayStatusBar
  Debug.Print "Calculation: " &amp; Application.Calculation
  Debug.Print "EnableEvents: " &amp; Application.EnableEvents

  ExcelValues.DisableExcelDisplayUpdates

  Debug.Print "During "
  Debug.Print "ScreenUpdating: " &amp; Application.ScreenUpdating
  Debug.Print "DisplayStatusBar: " &amp; Application.DisplayStatusBar
  Debug.Print "Calculation: " &amp; Application.Calculation
  Debug.Print "EnableEvents: " &amp; Application.EnableEvents

  MsgBox "Do your long running process here"

  ExcelValues.EnableExcelDisplayUpdates

  Debug.Print "After "
  Debug.Print "ScreenUpdating: " &amp; Application.ScreenUpdating
  Debug.Print "DisplayStatusBar: " &amp; Application.DisplayStatusBar
  Debug.Print "Calculation: " &amp; Application.Calculation
  Debug.Print "EnableEvents: " &amp; Application.EnableEvents

End Sub


*Note that the '@Folder("&lt;something&gt;") annotation is a feature of Rubberduck which is a great tool for helping to improve your VBA code. I'm an avid user and occasional contributor to the OSS project.